

# The Mass-Spring System

Math 308

This Maple session uses the mass-spring system to demonstrate the phase plane, direction fields, solution curves ("trajectories"), the extended phase space, and more.

The differential equation is

$$m \left( \frac{\partial^2}{\partial t^2} y(t) \right) + b \left( \frac{\partial}{\partial t} y(t) \right) + k y(t) = 0$$

(I've used b instead of the usual  $\gamma$  for the coefficient of friction.)

We convert this to a system by defining

$$v(t) = \frac{\partial}{\partial t} y(t)$$

Then

$$\frac{\partial}{\partial t} v(t) = -\frac{k y(t)}{m} - \frac{b \left( \frac{\partial}{\partial t} y(t) \right)}{m}$$

or

$$\frac{\partial}{\partial t} v(t) = -\frac{k y(t)}{m} - \frac{b v(t)}{m}$$

This equation along with the equation that defines v(t) make up a system of two equations.

```
> with(DEtools):
```

We first define the two equations that make up the system:

```
> eq1 := diff(y(t),t) = v(t);
```

$$eq1 := \frac{\partial}{\partial t} y(t) = v(t)$$

```
> eq2 := diff(v(t),t) = -(k/m)*y(t)-(b/m)*v(t);
```

$$eq2 := \frac{\partial}{\partial t} v(t) = -\frac{k y(t)}{m} - \frac{b v(t)}{m}$$

Now define a variable that holds the system:

```
> sys := [eq1,eq2];
```

$$\text{sys} := \left[ \frac{\partial}{\partial t} y(t) = v(t), \frac{\partial}{\partial t} v(t) = -\frac{k y(t)}{m} - \frac{b v(t)}{m} \right]$$


We substitute in some numerical values for the parameters, so we can plot some solutions.  
(The "subs" command makes the substitutions given by the first several arguments in the expression given as the last argument.)

```
> sys1 := subs(b=1/3,m=1,k=1,sys);
```

$$\text{sys1} := \left[ \frac{\partial}{\partial t} y(t) = v(t), \frac{\partial}{\partial t} v(t) = -y(t) - \frac{1}{3} v(t) \right]$$

Let's first look at the direction field. This is the same as the vector field, but all the vectors are plotted with the same length. We use the DEplot command.

```
> DEplot(sys1,[y(t),v(t)],t=0..1,y=-1..1,v=-1..1,scaling=constrained,
,color=black,title="Mass-Spring Direction Field (m=1,k=1,b=1/3)");
Mass-Spring Direction Field (m=1,k=1,b=1/3)
```




Let's add a solution curve, also called a *trajectory*. We will find the curve with initial condition  $y(0)=1$ ,  $v(0)=0$ . The ics argument to DEplot is a list of triples of the form  $[t_0, y_0, v_0]$ . Here we only put in one initial condition,  $[0,1,0]$ .

```
> ics := [[0,1,0]];
```

```
ics := [[0, 1, 0]]
```

```
> DEplot(sys1,[y(t),v(t)],t=-5..15,ics,y=-1..1,v=-1..1,scaling=const
rained,color=black,title="Mass-Spring Direction Field
(m=1,k=1,b=1/3)",linecolor=blue,stepsize=.1);
```


Mass-Spring Direction Field (m=1,k=1,b=1/3)



Let's plot  $y(t)$  vs.  $t$  for the same solution. The option "scene=[var1,var2]" let's us choose how the solution is plotted. If we say "scene=[t,y]", we get a plot of  $y(t)$  vs.  $t$ .

```
> DEplot(sys1,[y(t),v(t)],t=-5..15,ics,y=-2.5..1.5,v=-1..1,color=bla
ck,title="Mass-Spring
(m=1,k=1,b=1/3)",linecolor=blue,stepsize=.1,scene=[t,y]);
```


Mass-Spring (m=1,k=1,b=1/3)



Now let's see this solution in the extended phase space. Here we use the command **DEplot3d**. (Don't forget that you can click-and-drag in the 3D plot to see different views.)

```
> DEplot3d(sys1,[y(t),v(t)],t=-5..15,ics,y=-2.5..1.5,v=-1..1,color=black,title="Mass-Spring Extend Phase Space (m=1,k=1,b=1/3)",linecolor=blue,stepsize=.1,scene=[t,y,v],axes=normal);
```

Mass-Spring Extend Phase Space (m=1,k=1,b=1/3)



[ >